ELSEVIER

Mathematics and Computers in Simulation 49 (1999) 221–234

MATHEMATICS AND COMPUTERS IN SIMULATION

The diatomic anharmonic oscillator according to matrix mechanics[1]

M.B. Monagan*, J.F. Ogilvie

Center for Experimental and Constructive Mathematics, Simon Fraser University, Burnaby, BC, Canada V5A 1S6

Abstract

We determine the energies of states of the diatomic anharmonic oscillator by matrix mechanics in its original form as developed by Heisenberg, Born and Jordan using perturbation theory in successive orders. We present exact formulae for the second-, fourth-, and sixth-order contributions to the energy that were computed with Maple. The calculations involve matrices of finite rank with symbolic entries. We include the Maple programs. © 1999 IMACS/Elsevier Science B.V. All rights reserved.

Keywords: Anharmonic oscillator; Matrix mechanics; Maple programs

1. Introduction

Pioneer quantum mechanics [7] consists of matrix mechanics [1,2,5] and wave mechanics [8]. Both are based [10] on the following commutation law:

$$p_m q_n - q_m p_n = -\frac{\mathrm{i}h\delta_{m,n}}{2\pi}.$$

Two natural algebraic objects that have this non-commutative property are differential operators and matrices. Wave mechanics is based on differential operators such that p_n is generally replaced by:

$$-\mathrm{i}h\left(\frac{\partial}{\partial q_n}\cdot\right)\Big/2\pi.$$

The starting point for any wave-mechanical calculation is Schrödinger's equation, independent of time, that has the form [8]:

$$H\psi_k = E_k\psi_k$$

* Corresponding author. E-mail: monagan@cecm.sfu.ca

[1] Expanded version of a talk presented at the Applied Computer Algebra meeting in Maui, August 1997. This work was supported by NSERC of Canada.

PII: S0378-4754(99)00046-4

in which H is a Hamiltonian operator, and ψ_k is a wave function for state k. E_k is the total energy of the system in state k, which is the quantity of interest. In typical applications, the Hamiltonian operator consists of two contributions, potential energy $V(q_n)$ and kinetic energy $p_m^2/2\mu$ with μ the effective mass of the system. With a known Hamiltonian for a particular physical problem (e.g. for a diatomic oscillator such as HCl), a solution of the problem requires one to find ψ_k and eigenvalues E_k with certain boundary conditions. For simple systems one can find analytic solutions, whereas for more complicated systems perturbation techniques and numerical methods are the norm.

Before Schrödinger [8], in 1926, inspired by de Broglie's association of a wave with a moving particle, developed wave mechanics, Heisenberg [5] had already discovered that quantities describing physical systems obeyed the law of matrix multiplication. The first application of matrix mechanics was a linear harmonic oscillator, for which the Hamiltonian matrix is diagonal with diagonal elements being the discrete energies E_k. With Jordan's mathematical expertise, Heisenberg and Born [1,2] developed matrix mechanics into a fully fledged physical theory. First Schrödinger [9], then others, proved that the two approaches to quantum mechanics are exactly equivalent, that is, the energies E_k computed with either method must be the same. However, because physicists were more familiar with differential equations than with matrix algebra, when Schrödinger announced his theory physicists adopted the wave-mechanical approach and interest in matrix mechanics declined.

Our objectives in this paper are to show how matrix mechanics can be applied to a significant physical problem, namely a vibrating diatomic molecule modeled as an anharmonic oscillator, and to demonstrate how symbolic computation can extend to higher orders exact calculations done previously by hand. In comparison with wave mechanics, the level of mathematics required, namely elementary linear algebra, is within the grasp of university freshmen. In contrast, wave mechanics requires maturity in partial differential equations. The anharmonic oscillator cannot be solved exactly; hence in an approximate treatment we apply perturbation theory and develop the solution as a series in λ.

2. Matrix mechanics with perturbation theory

In matrix mechanics, every physical quantity is represented by a matrix [1], which are here of infinite rank. The general theory [10] to solve a problem in matrix mechanics is to transform, by a unitary transformation U, the Hamiltonian matrix H into diagonal form E, as expressed in this matrix equation:

$$UH = EU.$$

Here, for the harmonic oscillator, because the matrix H is diagonal U is the identity matrix; thus $E = H$. As for the anharmonic oscillator H is no longer diagonal, the solution of the problem requires explicit diagonalization. Because we employ perturbation theory [10], we develop each matrix as a series in a parameter λ. The term of order zero captures the harmonic oscillator. Thus:

$$H = H^{(0)} + \lambda H^{(1)} + \lambda^2 H^{(2)} + \cdots$$

$$U = U^{(0)} + \lambda U^{(1)} + \lambda^2 U^{(2)} + \cdots$$

$$E = E^{(0)} + \lambda E^{(1)} + \lambda^2 E^{(2)} + \cdots$$

The matrix H is a known quantity; it is determined by the nature of the physical problem that is in our application, a diatomic anharmonic oscillator in one dimension. The E matrix must be diagonal. The U matrix is unitary; thus $U^{\dagger}U = 1$. Thus to first order:

$$(U^{(0)} + \lambda U^{(1)})(H^{(0)} + \lambda H^{(1)}) = (E^{(0)} + \lambda E^{(1)})(U^{(0)} + \lambda U^{(1)}).$$

Expanding this to first order yields:

$$U^{(0)}H^{(0)} + (U^{(0)}H^{(1)} + U^{(1)}H^{(0)})\lambda + \mathrm{O}(\lambda^2) = E^{(0)}U^{(0)} + (E^{(0)}U^{(1)} + E^{(1)}U^{(0)})\lambda + \mathrm{O}(\lambda^2).$$

The solution for $E^{(0)}$ must satisfy:

$$U^{(0)}H^{(0)} = E^{(0)}U^{(0)}.$$

As $H^{(0)}$ is diagonal, we have that $E^{(0)} = H^{(0)}$ and $U^{(0)}$ is the identity matrix. Replacing $U^{(0)}$ by the identity matrix in the series we obtain this equation for the coefficient in λ:

$$U^{(1)}H^{(0)} + H^{(1)} = E^{(0)}U^{(1)} + E^{(1)}.$$

Here $E^{(0)}$, $H^{(0)}$, $H^{(1)}$ are known quantities and $E^{(1)}$ must be diagonal. Because $H^{(0)} = E^{(0)}$, and $H^{(0)}$ is diagonal, the diagonal entries of $U^{(1)}H^{(0)}$ are the same as $E^{(0)}U^{(1)}$; hence the diagonal entries of $E^{(1)}$ are equal those of $H^{(1)}$, which happen to be zero. For off-diagonal entries of the matrix U, one can also solve directly. One obtains linear equations:

$$(H^{(0)})_{j,j}(U^{(1)})_{i,j} + (H^{(1)})_{i,j} = (U^{(1)})_{i,j}(H^{(0)})_{i,i}.$$

The diagonal entries of $U^{(1)}$ are determined from a condition $U^{\mathrm{T}}U{=}1$ as follows:

$$(U^{(0)} + \lambda U^{(1)} + \lambda^2 U^{(2)} + \cdots)^{\mathrm{T}}(U^{(0)} + \lambda U^{(1)} + \lambda^2 U^{(2)} + \cdots) = 1,$$

which, to second order, is:

$$(U^{(0)})^{\mathrm{T}}U^{(0)} + ((U^{(0)})^{\mathrm{T}}U^{(1)} + (U^{(1)})^{\mathrm{T}}U^{(0)})\lambda + \mathrm{O}(\lambda^2) = 1.$$

As $U^{(0)}$ is the identity matrix, the coefficient of λ simplifies to:

$$U^{(1)} + (U^{(1)})^{\mathrm{T}} = 0.$$

This condition requires diagonal entries to be zero. The approach taken here of solving first for diagonal entries of E and off-diagonal entries of U, then for diagonal entries of U using the unitary condition, holds equally for terms of higher order. We note explicitly in calculations that follow how this method works for the next order. In summary, given H, U, E to order $K-1$, and $H^{(K)}$, we are able to calculate $E^{(K)}$ and $U^{(K)}$. To complete the method we here define H for the diatomic anharmonic oscillator.

A general Hamiltonian for a system of effective mass μ in one dimension is:

$$H = \frac{p^2}{2\mu} + V(x).$$

For an anharmonic oscillator we express the potential energy as

$$V(x) = k_2x^2 + k_3x^3 + k_4x^4 + \cdots.$$

In this case

$$H^{(0)} = \frac{p^2}{2\mu} + k_2x^2, \quad H^{(1)} = k_3x^3, \quad H^{(2)} = k_4x^4, \ldots$$

The displacement coordinate x becomes a matrix X in subsequent calculations. Elements of X are defined in a solution for the harmonic oscillator that we take as a starting point for this calculation. The matrix X is symmetric with two non-zero bands given by $X_{m,m+1} = A\sqrt{m+1}$ for $m = 0, 1, 2, \ldots,$ in which $A = \sqrt{h/2\mu\omega_0}$, h is Planck's constant, and ω_0 is the harmonic frequency. The matrix $H^{(0)}$ is equal to $E^{(0)}$, which is the solution for the harmonic oscillator; the diagonal matrix has diagonal entries $H^{(0)}_{m,m} = (m + 1/2)\omega$ for $m = 0, 1, 2, \ldots,$ in which $\omega = \omega_0 h$. This condition defines a sequence $H^{(1)}$, $H^{(2)}, \ldots$

3. Computations

To calculate terms of order K in a series for the matrices E and U, we proceed as follows. First we construct the matrix $H^{(K)}$ to a given rank. Next we create a matrix equation for the coefficient of λ^K by expanding the equation $UH = EU$ symbolically as a power series in λ. We insert all known matrices truncated to a specified rank and solve this equation for $E^{(K)}$ and off-diagonal entries of $U^{(K)}$. Then, to solve for diagonal entries of $U^{(K)}$, we employ an equation $U^{\mathrm{T}}U = 1$. We expand it as a power series in λ symbolically and select the coefficient of λ^K. We insert all known quantities calculated up to this point and solve for diagonal entries of $U^{(K)}$. After solving for $E^{(K)}$ and $U^{(K)}$, one can proceed to calculate terms of order $K + 1$.

3.1. Setting up Maple

The following statement tells Maple that λ is a scalar quantity:

```
> restart; constants := constants, lambda;
```

$$constants := false, \gamma, \infty, true, Catalan, FAIL, \pi, \lambda$$

The following Maple program simplifies matrices of formulae; it factors out any common factor present in each entry of a matrix

```
factormatrix := proc(A) local multigcd, B, L, N, D, G;
  multigcd := proc(a)
  local i, g;
    g := a[1];
    for i from 2 to nops(a) while g <> 1 do
      g := frontend(gcd, [g, a[i]]);
    od;
```

```
    g
  end;
  B := map(factor,A);
  L := op(map(op,convert(B,listlist))) minus {0};
  if nops(L) = 0 then RETURN(eval(B)) fi;
  N := multigcd(map(numer,L));
  if N <> 0 then N := N/icontent(N) fi;
  if N = 0 then RETURN(eval(B)) fi;
  D := multigcd(map(denom,L));
  if D <> 0 then D := D/icontent(D) fi;
  if D = 0 then RETURN(eval(B)) fi;
  G := N/D;
  if G = 1 or G = -1 then RETURN(eval(B)) fi;
  G * map(unapply(x/G,x),B)
end :
```

The following command is used to extract a submatrix from a matrix.

```
> with(linalg,submatrix);
```

[*submatrix*]

3.2. Constructing the matrix equations

The construction of these two matrix equations is done in the following Maple program that assigns coefficients of λ_i to variables G_i and UC_i up to a specified order $i = D$. This part of the computation is truly "symbolic" in that equations are in terms of unknown matrices. Later we fix the rank of these matrices, fill in known entries and solve for unknown entries.

```
makeEquations := proc(D) local i,H,U,E,G,Ut,UC;
  H := add(H.i * lambda^i, i = 0..D);
  U := 1 + add(U.i * lambda^i, i = 1..D);
  E := H0 + add(E.i * lambda^i, i = 1..D);
  G := expand(U& * H - E& * U);
  G := algsubs(lambda^(D + 1) = 0,G);
  Ut := 1 + add(transpose(U.i) * lambda^i, i = 1..D);
  UC := expand(U& * Ut - 1);
  UC := algsubs(lambda^(D + 1) = 0,UC);
  for i from 1 to D do
    G.i := coeff(G, lambda, i);
    UC.i := coeff(UC, lambda, i);
  od;
  RETURN();
end :
> makeEquations(3);
```

For instance, here are linear coefficients as determined previously.

```
> G1; UC1;
```

$$H1 - E1 + (U1\& * H0) - (H0\& * U1),$$

$$\text{transpose}(U1) + U1$$

and here are quadratic coefficients:

```
> G2; UC2;
```

$$-(E1\& * U1) + H2 + (U1\& * H1) + (U2\& * H0) - E2 - (H0\& * U2),$$

$$\text{transpose}(U2) + U2 + (U1\& * \text{transpose}(U1)).$$

That the matrix $H^{(0)}$ is diagonal makes the algorithm work so simply. In the linear coefficient above, the quantity $U^{(1)}H^{(0)} - H^{(0)}U^{(1)}$ has diagonal entries that are zero; therefore diagonal entries of $E^{(1)}$ are equal to those in $H^{(1)}$. In the quadratic coefficient, the unknown quantities are $U^{(2)}H^{(0)} - H^{(0)}U^{(2)}$ and $E^{(2)}$. Diagonal entries of the first expression are again zero; hence diagonal entries of $E^{(2)}$ are given by diagonal entries of a matrix $H^{(2)} + U^{(1)}H^{(1)} - E^{(1)}U^{(1)}$. The same observation applies for terms of greater order that have a general form $E^{(K)} + U^{(K)}H^{(0)} - H^{(0)}U^{(K)} + \cdots = 0$. Diagonal entries of $E^{(K)}$ are simply determined and off-diagonal entries of $U^{(K)}$ yield simple linear equations. Here is the cubic coefficient.

```
> G3; UC3;
```

$$(U1\& * H2) + (U3\& * H0) - E3 + (U2\& * H1) - (H0\& * U3) + H3 - (E1\& * U2) - (E2\& * U1),$$

$$(U1\& * \text{transpose}(U2)) + \text{transpose}(U3) + (U2\& * \text{transpose}(U1)) + U3.$$

3.3. Constructing the H matrices

The following program constructs the matrix $H^{(K)}$ for $K > 0$. Recall that all matrices have formally infinite rank. A parameter N specifies the rank of a truncated matrix. To obtain $H^{(K)}$ correct to N rows and N columns we must compute $X^{(K+2)}$ and we need to begin with $N + K$ rows and columns for X.

```
makeH := proc(K,N) local m,X,H;
X := matrix(N+K,N+K,0);
for m from 0 to N+K-2 do
  X[m+1,m+2] := A*sqrt(m+1);
  X[m+2,m+1] := X[m+1,m+2];
od;
H := evalm(k.(K+2)*X^(K+2));
submatrix(H,1..N,1..N);
end :
```

Here is the matrix $H^{(1)}$ truncated to rank 6.

```
> H1 := makeH(1,6) : factormatrix(H1);
```

$$k3A^3 \begin{bmatrix} 0 & 3 & 0 & \sqrt{2}\sqrt{3} & 0 & 0 \\ 3 & 0 & 6\sqrt{2} & 0 & 2\sqrt{2}\sqrt{3} & 0 \\ 0 & 6\sqrt{2} & 0 & 9\sqrt{3} & 0 & 2\sqrt{3}\sqrt{5} \\ \sqrt{2}\sqrt{3} & 0 & 9\sqrt{3} & 0 & 24 & 0 \\ 0 & 2\sqrt{2}\sqrt{3} & 0 & 24 & 0 & 15\sqrt{5} \\ 0 & 0 & 2\sqrt{3}\sqrt{5} & 0 & 15\sqrt{5} & 0 \end{bmatrix}$$

Formula for bands can be obtained by interpolation. We have:

$$\frac{H1_{m,m+1}}{A^3\,k3} = 3m\sqrt{m+1} \quad \text{for } 0 \le m$$

and

$$\frac{H1_{m,m+3}}{A^3\,k3} = \sqrt{(m+1)(m+2)(m+3)} \quad \text{for } 0 \le m.$$

Likewise, here is the matrix $H^{(2)}$ truncated to rank 6.

```
> H2 := makeH(2,6) : factormatrix(H2);
```

$$k4A^4 \begin{bmatrix} 3 & 0 & 6\sqrt{2} & 0 & 2\sqrt{2}\sqrt{3} & 0 \\ 0 & 15 & 0 & 10\sqrt{2}\sqrt{3} & 0 & 2\sqrt{2}\sqrt{3}\sqrt{5} \\ 6\sqrt{2} & 0 & 39 & 0 & 28\sqrt{3} & 0 \\ 0 & 10\sqrt{2}\sqrt{3} & 0 & 75 & 0 & 36\sqrt{5} \\ 2\sqrt{2}\sqrt{3} & 0 & 28\sqrt{3} & 0 & 123 & 0 \\ 0 & 2\sqrt{2}\sqrt{3}\sqrt{5} & 0 & 36\sqrt{5} & 0 & 183 \end{bmatrix}$$

Diagonal entries are given by:

$$\frac{H2_{m,m}}{A^4k4} = 6m^2 + 6m + 3 \quad \text{for } 0 \le m$$

and the next band is given by:

$$\frac{H2_{m,m+2}}{A^4k4} = 2(2m+3)\sqrt{(m+1)(m+2)} \quad \text{for } 0 \le m$$

and the outermost band by:

$$\frac{H2_{m,m+4}}{A^4k2} = \sqrt{(m+1)(m+2)(m+3)(m+4)} \quad \text{for } 0 \le m.$$

In general matrices H are symmetric and banded with $K+3$ non-zero bands and bands of zeroes between non-zero bands. Formulae for bands can be derived on inspection and interpolation.

3.4. Solving for E^K and U^K

The following program calculates matrices E and U to order K truncated to rank N.

```
AnHarmonic := proc(D,N)
  global U0,H0,E0,Order;
```

```
local i,j,m,x,K,T,Z,U,E,H,eqns,vars,sols;
makeEquations(D);
U0 := array(1..N,1..N,identity);
H0 := array(1..N,1..N,diagonal);
for m from 0 to N-1 do H0[m+1,m+1] := (m+1/2)*omega od :
E0 := H0;
for K from 1 to D do
  print(`Solving for the coefficient of`, lambda^K);
  H.K := makeH(K,N);
  E.K := array(1..N,1..N,diagonal);
  U.K := array(1..N,1..N);
  U := evaln(U.K); E := evaln(E.K); H := evaln(H.K);
  printf(`The matrix %s\n`,H); print(factormatrix(H.K));
  printf(`The matrix equation to be solved\n`); print(G.K);
  Z := evalm(G.K);
  if K=1 then
    printf(`The expanded matrix equation to be solved\n`);
    print(Z) fi;
  eqns := normal(convert(Z,set));
  vars := {seq(seq(U[i,j],j=1..N),i=1..N), seq(E[i,i],i=1..N)} minus
          {seq(U[i,i],i=1..N)};
  sols := solve(eqns,vars); assign(sols);
  printf(`The matrix %s\n`,E); print(factormatrix(E));
  T := evalm(UC.K);
  printf(`Condition for the next U matrix`); print(UC.K=0);
  eqns := normal(convert(T,set));
  vars := {seq(U[i,i],i=1..N)};
  sols := solve(eqns,vars); assign(sols);
  printf(`The matrix %s\n`,U);print(factormatrix(U));
 od;
end :
```

We calculate and display $H^{(1)}$, $E^{(1)}$, $U^{(1)}$, then $H^{(2)}$, $E^{(2)}$, $U^{(2)}$. We also display the symbolic matrix equations. Note that $E^{(1)} = 0$, and all E matrices of odd order are 0.

```
> AnHarmonic(2,6);
```

Solving for the coefficient of, λ

The matrix H1

$$k3A^3\begin{bmatrix} 0 & 3 & 0 & \sqrt{2}\sqrt{3} & 0 & 0 \\ 3 & 0 & 6\sqrt{2} & 0 & 2\sqrt{2}\sqrt{3} & 0 \\ 0 & 6\sqrt{2} & 0 & 9\sqrt{3} & 0 & 2\sqrt{3}\sqrt{5} \\ \sqrt{2}\sqrt{3} & 0 & 9\sqrt{3} & 0 & 24 & 0 \\ 0 & 2\sqrt{2}\sqrt{3} & 0 & 24 & 0 & 15\sqrt{5} \\ 0 & 0 & 2\sqrt{3}\sqrt{5} & 0 & 15\sqrt{5} & 0 \end{bmatrix}$$

The matrix equation to be solved:

$$H1 - E1 + (U1\& * H0) - (H0\& * U1)$$

The expanded matrix equation to be solved:

$$[-E1_{1,1}, 3k3A^3 + U1_{1,2}\omega, 2U1_{1,3}\omega, \%1 + 3U1_{1,4}\omega, 4U1_{1,5}\omega, 5U1_{1,6}\omega]$$

$$[3k3A^3 - U1_{2,1}\omega, -E1_{2,2}, 6k3A^3\sqrt{2} + U1_{2,3}\omega, 2U1_{2,4}\omega, 2\%1 + 3U1_{2,5}\omega, 4U1_{2,6}\omega]$$

$$[-2U1_{3,1}\omega, 6k3A^3\sqrt{2} - U1_{3,2}\omega, -E1_{3,3}, 9k3A^3\sqrt{3} + U1_{3,4}\omega, 2U1_{3,5}\omega, 2k3A^3\sqrt{3}\sqrt{5} + 3U1_{3,6}\omega]$$

$$[\%1 - 3U1_{4,1}\omega, -2U1_{4,2}\omega, 9k3A^3\sqrt{3} - U1_{4,3}\omega, -E1_{4,4}, 24k3A^3 + U1_{4,5}\omega, 2U1_{4,6}\omega]$$

$$[-4U1_{5,1}\omega, 2\%1 - 3U1_{5,2}\omega, -2U1_{5,3}\omega, 24k3A^3 - U1_{5,4}\omega, -E1_{5,5}, 15k3A^3\sqrt{5} + U1_{5,6}\omega]$$

$$[-5U1_{6,1}\omega, -4U1_{6,2}\omega, 2k3A^3\sqrt{3}\sqrt{5} - 3U1_{6,3}\omega, -2U1_{6,4}\omega, 15k3A^3\sqrt{5} - U1_{6,5}\omega, -E1_{6,6}]$$

$$\%1 := k3A^3\sqrt{2}\sqrt{3}$$

The matrix E1

$$\begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

The condition for the next U matrix is:

$$\text{transpose}(U1) + U1 = 0.$$

The matrix U1

$$k3A^3\omega^{-1}\begin{bmatrix} 0 & -3 & 0 & -\frac{1}{3}\sqrt{2}\sqrt{3} & 0 & 0 \\ 3 & 0 & -6\sqrt{2} & 0 & -\frac{2}{3}\sqrt{2}\sqrt{3} & 0 \\ 0 & 6\sqrt{2} & 0 & -9\sqrt{3} & 0 & -\frac{2}{3}\sqrt{3}\sqrt{5} \\ \frac{1}{3}\sqrt{2}\sqrt{3} & 0 & 9\sqrt{3} & 0 & -24 & 0 \\ 0 & \frac{2}{3}\sqrt{2}\sqrt{3} & 0 & 24 & 0 & -15\sqrt{5} \\ 0 & 0 & \frac{2}{3}\sqrt{3}\sqrt{5} & 0 & 15\sqrt{5} & 0 \end{bmatrix}$$

Solving for the coefficient of, λ^2

The matrix H2

$$k4A^4\begin{bmatrix} 3 & 0 & 6\sqrt{2} & 0 & 2\sqrt{2}\sqrt{3} & 0 \\ 0 & 15 & 0 & 10\sqrt{2}\sqrt{3} & 0 & 2\sqrt{2}\sqrt{3}\sqrt{5} \\ 6\sqrt{2} & 0 & 39 & 0 & 28\sqrt{3} & 0 \\ 0 & 10\sqrt{2}\sqrt{3} & 0 & 75 & 0 & 36\sqrt{5} \\ 2\sqrt{2}\sqrt{3} & 0 & 28\sqrt{3} & 0 & 123 & 0 \\ 0 & 2\sqrt{2}\sqrt{3}\sqrt{5} & 0 & 36\sqrt{5} & 0 & 183 \end{bmatrix}$$

The matrix equation to be solved:

$$-(E1\& * U1) + H2 + (U1\& * H1) + (U2\& * H0) - E2 - (H0\& * U2)$$

The matrix E2

$$A^4\omega^{-1}\begin{bmatrix} 3k4\omega - 11k3^2A^2, 0, 0, 0, 0, 0 \\ 0, 15k4\omega - 71k3^2A^2, 0, 0, 0, 0 \\ 0, 0, 39k4\omega - 191k3^2A^2, 0, 0, 0 \\ 0, 0, 0, 75k4\omega - 331k3^2A^2, 0, 0 \\ 0, 0, 0, 0, 123k4\omega - 541k3^2A^2, 0 \\ 0, 0, 0, 0, 0, 183k4\omega + 1145k3^2A^2 \end{bmatrix}$$

The condition for the next U matrix:

$$\text{transpose}(U2) + U2 + (U1\& * \text{transpose}(U1)) = 0.$$

The matrix U2

$$A^4\omega^{-2}\left[-\frac{29}{6}k3^2A^2, 0, \frac{3}{2}(-2k4\omega + 9k3^2A^2)\sqrt{2}, 0, \frac{1}{2}(-k4\omega + 7k3^2A^2)\sqrt{2}\sqrt{3}, 0\right]$$

$$\left[0, -\frac{251}{6}k3^2A^2, 0, \frac{1}{2}(-10k4\omega + 67k3^2A^2)\sqrt{2}\sqrt{3}, 0, \frac{1}{2}\sqrt{2}\sqrt{3}\sqrt{5}(-k4\omega + 11k3^2A^2)\right]$$

$$\left[-\frac{3}{2}(-2k4\omega + 3k3^2A^2)\sqrt{2}, 0, -\frac{965}{6}k3^2A^2, 0, (-14k4\omega + 121k3^2A^2)\sqrt{3}, 0\right]$$

$$\left[0, \frac{1}{2}(10k4\omega + 7k3^2A^2)\sqrt{2}\sqrt{3}, 0, -\frac{2459}{6}k3^2A^2, 0, 9(-2k4\omega + 17k3^2A^2)\sqrt{5}\right]$$

$$\left[\frac{1}{2}(k4\omega + 13k3^2A^2)\sqrt{2}\sqrt{3}, 0, (14k4\omega + 37k3^2A^2)\sqrt{3}, 0, -\frac{5111}{6}k3^2A^2, 0\right]$$

$$\left[0, \frac{1}{2}\sqrt{2}\sqrt{3}\sqrt{5}(k4\omega + 17k3^2A^2), 0, 9(2k4\omega + 21k3^2A^2)\sqrt{5}, 0, -\frac{3395}{6}k3^2A^2\right]$$

To verify that all computations performed with the program AnHarmonic are correct, that is, that we have correctly diagonalized H to $O(\lambda^3)$, we test that the equation $UH = EU$ holds to $O(\lambda^3)$.

```
> eqn := (1 + add(lambda^i * U.i, i = 1..2))& * add(lambda^i * H.i, i = 0..2)
>= add(lambda^i * E.i, i = 0..2)& * (1 + add(lambda^i * U.i, i = 1..2));
```

$$eqn := (1+U1\lambda+U2\lambda^2)\&*(H0+H1\lambda+H2\lambda^2)=(H0+E1\lambda+E2\lambda^2)\&*(1+U1\lambda+U2\lambda^2)$$

```
> map(simplify@series, evalm(lhs(eqn) - rhs(eqn)), lambda, 3);
```

$$\begin{bmatrix} O(\lambda^4) & O(\lambda^3) & O(\lambda^4) & O(\lambda^3) & O(\lambda^4) & O(\lambda^3) \\ O(\lambda^3) & O(\lambda^4) & O(\lambda^3) & O(\lambda^4) & O(\lambda^3) & O(\lambda^4) \\ O(\lambda^4) & O(\lambda^3) & O(\lambda^4) & O(\lambda^3) & O(\lambda^4) & O(\lambda^3) \\ O(\lambda^3) & O(\lambda^4) & O(\lambda^3) & O(\lambda^4) & O(\lambda^3) & O(\lambda^4) \\ O(\lambda^4) & O(\lambda^3) & O(\lambda^4) & O(\lambda^3) & O(\lambda^4) & O(\lambda^3) \\ O(\lambda^3) & O(\lambda^4) & O(\lambda^3) & O(\lambda^4) & O(\lambda^3) & O(\lambda^4) \end{bmatrix}$$

We proceed to the second-order contribution to the energy; it is the

```
> map(expand,E2);
```

$$\begin{bmatrix} 3\ k4\ A^4 - 11(k3^2A^6/\omega), 0, 0, 0, 0, 0 \\ 0, 15\ k4\ A^4 - 71(k3^2A^6/\omega), 0, 0, 0, 0 \\ 0, 0, 39\ k4\ A^4 - 191(k3^2A^6/\omega), 0, 0, 0 \\ 0, 0, 0, 75\ k4\ A^4 - 331(k3^2A^6/\omega), 0, 0 \\ 0, 0, 0, 0, 123\ k4\ A^4 - 541(k3^2A^6/\omega), 0 \\ 0, 0, 0, 0, 0, 183\ k4\ A^4 + 1145(k3^2A^6/\omega) \end{bmatrix}$$

Because we truncated the H matrices, not all entries in matrices E and U are correct. In fact only the first three diagonal entries in $E^{(2)}$ are correct. We compare our results with those of Wu [10]; he employed the following definition for H:

$$H = \frac{p^2}{2\mu} + \frac{k_1x^2}{2!} + \frac{k_2x^3}{3!} + \frac{k_3x^4}{4!} + \cdots$$

We convert our results for $E^{(2)}$ to conform to Wu's notation:

```
> W2 := subs(k3 = k2/3!, k4 = k3/4!,
> omega = omega[0] * h, submatrix(%, 1..3, 1..3));
```

$$W2 := \begin{bmatrix} (1/8)k3A^4 - (11/36)(k2^2A^6/\omega_0 h) & 0 & 0 \\ 0 & (5/8)k3A^4 - (71/36)(k2^2A^6/\omega_0 h) & 0 \\ 0 & 0 & (13/8)k3A^4 - (191/36)(k2^2A^6/\omega_0 h) \end{bmatrix}$$

These diagonal entries are consistent with Eq. (II-140) in [10]. The correction to the energy of the harmonic oscillator in Wu's equation we obtain by interpolation from the three values in the above matrix. Using Maple, we obtain:

```
> interp([0, 1, 2], [W2[1, 1], W2[2, 2], W2[3, 3]], n);
```

$$-\frac{1}{72}\frac{A^4(60k2^2A^2 - 18k3\omega_0 h)n^2}{\omega_0 h} - \frac{1}{72}\frac{A^4(60k2^2A^2 - 18k3\omega_0 h)n}{\omega_0 h} - \frac{1}{72}\frac{A^4(-9k3\omega_0 h + 22k2^2A^2)}{\omega_0 h}$$

Two possible ways to write this expression are as a polynomial in A or as a polynomial in $n+\frac{1}{2}$. In Maple:

```
> collect(%,A,factor);
```

$$-\frac{1}{36}\frac{k2^2(30n^2+30n+11)A^6}{\omega_0 h}+\frac{1}{8}k3(2n^2+2n+1)A^4$$

```
> map(collect,series(%,n=-1/2),A);
```

$$\left(-\frac{7}{72}\frac{k2^2A^6}{\omega_0 h}+\frac{1}{16}k3A^4\right)+\left(-\frac{5}{6}\frac{k2^2A^6}{\omega_0 h}+\frac{1}{4}k3A^4\right)\left(n+\frac{1}{2}\right)^2$$

Corrections to higher order are obtained on running program AnHarmonic with matrices of sufficient rank. We present in Appendix A results for correction terms $E^{(4)}$ and $E^{(6)}$.

We end this section with comments on computations, first, regarding the construction of $H^{(K)}$ for $K>0$. When we first computed these matrices, we used $H^{(1)}_{i,j}=k_3(\sum_{i,j}x_{m,i}x_{i,j}x_{j,n})$ and $H^{(2)}_{i,j}=k_4(\sum_{i,j,k}x_{m,i}x_{i,j}x_{j,k}x_{k,n})$ etc. This procedure yields an algorithm to compute $H^{(K)}$, which takes time exponential in K. The algorithm based on computing the matrix power X^K is superior; it is rapid enough that there is little point in using formulae for the bands of $H^{(K)}$.

The linear system of equations that is constructed and solved is a diagonal system, but because the formulae involve several square roots of small integers, Maple runs much more slowly than if the coefficients were simple integers. Part of the reason for this is that Maple takes care to determine algebraic relationships between square roots, e.g. $\sqrt{6}-\sqrt{2}\sqrt{3}=0$. In our case, if one does not find $\sqrt{2}$, $\sqrt{3}$ and $\sqrt{6}$ objectionable in the answer, one can accelerate calculations by a factor of 10 or more if one solves the equations ''blindly''. We have written our own solver to do this.

4. Conclusion

We demonstrate that matrix mechanics can be used effectively to solve a significant problem in quantum mechanics with perturbation theory. We apply matrix mechanics in its purest form, not simply an expression of wave mechanics in matrix form [4]. Moreover, we apply a perturbative approach whereby successive terms in potential energy expressed as a power series are treated in successive orders of the theory, not with the entire set of terms beyond a quadratic term in potential energy taken as a single perturbation to only first order as in alternative applications of perturbation theory. Because the matrix $H^{(0)}$ is diagonal for a diatomic harmonic oscillator, the requisite matrix algebra is simple. Computations can consequently be made exactly and analytically to high order using a symbolic processor such as Maple, even though perturbation theory is an intrinsically approximate method. Alternative analytic approaches to solutions of eigenenergies of a diatomic anharmonic oscillator involve a BKW or quasi-classical method and hypervirial perturbation theory in wave mechanics. We have proved that the results here can be readily converted into a form identical with those of the specified approaches [3,6], also programmed in Maple.

For reference we include in Appendix A explicit formulae for terms in energy of fourth order and sixth order. The Maple code for all calculations and results presented in the paper is available by electronic mail from the first author.

Appendix A

A. Correction terms of fourth and sixth order

```
> AnHarmonic(4,10):
> factormatrix(submatrix(E4,1..4,1..4));
> interp([0,1,2,3],[seq(E4[i,i],i=1..4)],n):
> collect(%,[A,k3,k4,k5,k6,omega],factor);
```

$$-30\frac{(2n+1)(47n^2+47n+31)k3^4A^{12}}{\omega^3}+36\frac{(2n+1)(25n^2+25n+19)k4k3^2A^{10}}{\omega^2}$$
$$+\left(-10\frac{(2n+1)(14n^2+14n+13)k5k3}{\omega}-2\frac{(2n+1)(17n^2+17n+21)k4^2}{\omega}\right)A^8$$
$$+5(2n+1)(2n^2+2n+3)k6A^6$$

```
> evaln(E4[n,n])=map(collect,taylor(%,n=-1/2),A,factor);
```

$$E4_{n,n}=\left(-1155\frac{k3^4A^{12}}{\omega^3}+918\frac{k3^2k4A^{10}}{\omega^2}-\frac{(190k5k3+67k4^2)A^8}{\omega}+25k6A^6\right)\left(n+\frac{1}{2}\right)$$
$$+\left(-2820\frac{k3^4A^{12}}{\omega^3}+1800\frac{k3^2k4A^{10}}{\omega^2}-4\frac{(70k5k3+17k4^2)A^8}{\omega}+20k6A^6\right)\left(n+\frac{1}{2}\right)^3$$

```
> AnHarmonic(6,18):
> interp([0,1,2,3,4],[seq(E6[i,i],i=1..5)],n):
> evaln(E6[n,n])=map(collect,taylor(%,n=-1/2),A,factor);
```

$$E6_{n,n}=\left(-\frac{101479}{4}\frac{k3^6A^{18}}{\omega^5}+\frac{131817}{4}\frac{k3^4k4A^{16}}{\omega^4}-\frac{1}{4}\frac{k3^2(29554k5k3+40261k4^2)A^{14}}{\omega^3}\right.$$
$$+\frac{1}{4}\frac{(11334k4k5k3+6055k6k3^2+1539k4^3)A^{12}}{\omega^2}-\frac{3}{8}\frac{(630k6k4+770k7k3+369k5^2)A^{10}}{\omega}$$
$$\left.+\frac{315}{8}k8A^8\right)+\left(-418110\frac{k3^6A^{18}}{\omega^5}+479970\frac{k3^4k4A^{16}}{\omega^4}\right.$$
$$-6\frac{k3^2(15910k5k3+20671k4^2)A^{14}}{\omega^3}+6\frac{(4890k4k5k3+2845k6k3^2+569k4^3)A^{12}}{\omega^2}$$
$$\left.-5\frac{(354k6k4+546k7k3+217k5^2)A^{10}}{\omega}+245k8A^8\right)\left(n+\frac{1}{2}\right)^2$$
$$+\left(-463020\frac{k3^6A^{18}}{\omega^5}+465300\frac{k3^4k4A^{16}}{\omega^4}-60\frac{k3^2(1302k5k3+1663k4^2)A^{14}}{\omega^3}\right.$$
$$+60\frac{(322k4k5k3+181k6k3^2+25k4^3)A^{12}}{\omega^2}-30\frac{(22k6k4+42k7k3+21k5^2)A^{10}}{\omega}$$
$$\left.+70k8A^8\right)\left(n+\frac{1}{2}\right)^4$$

References

[1] M. Born, W. Heisenberg, P. Jordan, On quantum mechanics II, Z. Phys. 35 (1926) 557–615.
[2] M. Born, P. Jordan, On quantum mechanics, Z. Phys. 34 (1925) 858–888.
[3] F. Fernandez, J. Ogilvie, Symbolic programs for analysis of diatomic vibration–rotational spectra, MapleTech. 5(1) (1998) 42–46.
[4] W. Gordy, R.L. Cook, Microwave Molecular Spectra, 3rd ed., cf. Appendix A, Wiley, New York, 1984.
[5] W. Heisenberg, Quantum-theoretical reinterpretation of kinematic and mechanical relations, Z. Phys. 33 (1925) 879–893.
[6] J. Ogilvie, M. Monagan, Symbolic computation in molecular spectroscopy, MapleTech. 4(2) (1998) 100–108.
[7] H. Primas, Chemistry, Quantum Mechanics and Reductionism, 2nd ed., Springer, Berlin, 1983.
[8] E. Schrodinger, Quantization as a problem of characteristic values I, Ann. Phys. Leipzig 79 (1926) 361–376.
[9] E. Schrodinger, The relation between quantum mechanics of Heisenberg, Born and Jordan and that of Schrodinger, Ann. Phys. Leipzig 79 (1926) 734–756.
[10] T.-Y. Wu, Quantum Mechanics, World Scientific, Singapore, 1986.